Gryph Programming Language Syntax in EBNF

Vitor Greati Artur Curinga Carlos Vieira Vinícius Campos

June 17, 2018

Contents

1 Syntax in EBNF

1.1 Program

$$
\begin{aligned}
\langle \text{program} \rangle &\models \langle \text{program-unit} \rangle \{ \langle \text{program-unit} \rangle \} \\
\langle \text{program-unit} \rangle &\models \langle \text{stmt} \rangle \mid \langle \text{subprog-def} \rangle \mid \langle \text{type-def} \rangle \mid \langle \text{include} \rangle \\
\langle \text{include} \rangle &\models \textbf{use}\ \langle \text{string-lit} \rangle
\end{aligned}
$$

1.2 Identifiers

$$
\begin{aligned}
\langle \text{id-list} \rangle &\models \langle \text{identifier} \rangle \{ \textbf{,} \langle \text{identifier} \rangle \} \\
\langle \text{identifier} \rangle &\models \langle \text{alpha} \rangle \langle \text{id-tail} \rangle \\
\langle \text{user-type-id} \rangle &\models \langle \text{upper-alpha} \rangle \langle \text{id-tail} \rangle \\
\langle \text{id-tail} \rangle &\models \{ \langle \text{alpha-num} \rangle \} \{ \textbf{'} \} \\
\langle \text{alpha-num} \rangle &\models \langle \text{alpha} \rangle \mid \langle \text{digit} \rangle \mid \textbf{\_} \\
\langle \text{digit} \rangle &\models \textbf{0} \mid \ldots \mid \textbf{9} \\
\langle \text{alpha} \rangle &\models \langle \text{upper-alpha} \rangle \mid (\textbf{a} \mid \ldots \mid \textbf{z}) \\
\langle \text{upper-alpha} \rangle &\models \textbf{A} \mid \ldots \mid \textbf{Z}
\end{aligned}
$$

1.3 Statements

⟨stmt-list⟩ ⊨ ⟨stmt⟩{⟨stmt⟩}
⟨stmt-block⟩ ⊨ **{** ⟨stmt-list⟩ **}**
⟨stmt⟩ ⊨ ⟨matched-stmt⟩ | ⟨unmatched-stmt⟩
⟨block-or-matched⟩ ⊨ ⟨stmt-block⟩ | ⟨matched-stmt⟩
⟨matched-stmt⟩ ⊨ ⟨matched-if-else⟩ | ⟨iteration-stmt⟩ | ⟨simple-stmt⟩
⟨unmatched-stmt⟩ ⊨ ⟨if-stmt⟩ | ⟨unmatched-if-else⟩
⟨simple-stmt⟩ ⊨ (⟨io-stmt⟩ | ⟨var-stmt⟩ | ⟨add-del-stmt⟩ | ⟨subprog-call⟩ | ⟨return-stmt⟩ | ⟨break-stmt⟩)**;**

1.3.1 IO

⟨io-stmt⟩ ⊨ ⟨read-stmt⟩ | ⟨write-stmt⟩
⟨read-stmt⟩ ⊨ **read** ⟨identifier⟩
⟨write-stmt⟩ ⊨ **print** ⟨expression⟩

1.3.2 Variables

⟨var-stmt⟩ ⊨ ⟨var-decl-stmt⟩ | ⟨var-attr-stmt⟩
⟨var-decl-list⟩ ⊨ ⟨var-decl-stmt⟩{**;**⟨var-decl-stmt⟩}**;**
⟨var-decl-stmt⟩ ⊨ ⟨id-list⟩**:**⟨type⟩[⟨var-attr⟩]
⟨var-attr-stmt⟩ ⊨ ⟨lhs-expr-list⟩⟨var-attr⟩
⟨id-attr⟩ ⊨ ⟨identifier⟩**=**⟨expression⟩
⟨id-attr-list⟩ ⊨ ⟨id-attr⟩{, ⟨id-attr⟩}
⟨var-attr⟩ ⊨ **=**⟨expr-list⟩

1.3.3 Insertion and removal

⟨add-del-stmt⟩ ⊨ ⟨add-stmt⟩ | ⟨del-stmt⟩
⟨add-stmt⟩ ⊨ **add** ⟨expression⟩ **in** ⟨lhs-expr⟩
⟨del-stmt⟩ ⊨ **del** ⟨expression⟩ **from** ⟨lhs-expr⟩

1.4 Control Structures

1.4.1 Conditionals

⟨if-expr⟩ ⊨ **if (**⟨expression⟩**)**
⟨if-stmt⟩ ⊨ ⟨if-expr⟩ ⟨stmt⟩
⟨unmatched-if-else⟩ ⊨ ⟨if-expr⟩ ⟨matched-stmt⟩ **else** ⟨unmatched-stmt⟩
⟨matched-if-else⟩ ⊨ ⟨if-expr⟩ ⟨block-or-matched⟩ **else** ⟨block-or-matched⟩ | ⟨if-expr⟩ ⟨stmt-block⟩

1.4.2 Iteration

⟨iteration-stmt⟩ ⊨ ⟨for-stmt⟩ | ⟨while-stmt⟩ | ⟨bfs-dfs-stmt⟩
⟨while-stmt⟩ ⊨ **while** ⟨expression⟩ ⟨block-or-matched⟩
⟨for-loop⟩ ⊨ **for** ⟨id-list⟩ **over** ⟨expr-list⟩
⟨for-stmt⟩ ⊨ ⟨for-loop⟩⟨block-or-matched⟩
⟨bfs-dfs-loop⟩ ⊨ (**bfs** | **dfs**) ⟨id-list⟩ **over** ⟨expr-list⟩
⟨bfs-dfs-stmt⟩ ⊨ ⟨dfs-dfs-loop⟩⟨block-or-matched⟩
⟨break-stmt⟩ ⊨ **break**

1.5 Subprograms

⟨subprog-def⟩ ⊨ **sub**⟨identifier⟩**(**⟨parameters⟩**)**[**:**⟨type⟩]⟨stmt-block⟩
⟨parameters⟩ ⊨ ⟨var-stmt⟩{**;**⟨var-stmt⟩}
⟨subprog-call⟩ ⊨ ⟨identifier⟩**(**[⟨arguments⟩]**)**
⟨arguments⟩ ⊨ ⟨id-attr-list⟩ | ⟨expr-list⟩
⟨return-stmt⟩ ⊨ **return**

1.6 Types

⟨type-list⟩ ⊨ ⟨type⟩{**,**⟨type⟩}
⟨type⟩ ⊨ ⟨native-type⟩ | ⟨user-type-id⟩
⟨native-type⟩ ⊨ ⟨primitive-type⟩ | ⟨composite-type⟩
⟨primitive-type⟩ ⊨ **int** | **float** | **char** | **string**
⟨composite-type⟩ ⊨ **[**⟨type⟩**]** | **|**⟨type⟩**|** | **(**⟨type⟩**,**⟨type-list⟩**)** | ⟨graph-type⟩
⟨graph-type⟩ ⊨ **<**⟨type⟩**>** | **<**⟨type⟩**,**⟨type⟩**>**
⟨type-def⟩ ⊨ ⟨user-type-id⟩**{**⟨var-decl-list⟩**}**

Observation Although there is no maximum size for tuples in the definition above, there may be one for specific language implementations.

1.7 Expressions

⟨expr-list⟩ ⊨ ⟨expression⟩{**,**⟨expression⟩}
⟨expression⟩ ⊨ ⟨logical-xor-expr⟩
⟨logical-xor-expr⟩ ⊨ ⟨logical-or-expr⟩{**xor** ⟨logical-or-expr⟩}
⟨logical-or-expr⟩ ⊨ ⟨logical-and-expr⟩{**or** ⟨logical-and-expr⟩}
⟨logical-and-expr⟩ ⊨ ⟨equality-expr⟩{**and** ⟨equality-expr⟩}
⟨equality-expr⟩ ⊨ ⟨rel-expr⟩{⟨equality-op⟩ ⟨rel-expr⟩}
⟨rel-expr⟩ ⊨ ⟨add-expr⟩{⟨rel-op⟩⟨add-expr⟩}
⟨add-expr⟩ ⊨ ⟨mult-expr⟩{⟨add-op⟩⟨mult-expr⟩}
⟨mult-expr⟩ ⊨ ⟨exp-expr⟩{⟨mult-op⟩⟨exp-expr⟩}
⟨exp-expr⟩ ⊨ ⟨cast-expr⟩[⟨exp-op⟩⟨exp-expr⟩]
⟨cast-expr⟩ ⊨ ⟨unary-expr⟩{**@**⟨type⟩}

⟨unary-expr⟩ ⊨ ⟨unary-op⟩⟨cast-expr⟩ | ⟨postfix-expr⟩
⟨postfix-expr⟩ ⊨ ⟨primary-expr⟩{⟨access-expr⟩}
⟨lhs-expr⟩ ⊨ ⟨identifier⟩{⟨access-expr⟩}
⟨lhs-expr-list⟩ ⊨ ⟨lhs-expr⟩{**,**⟨lhs-expr⟩}
⟨access-expr⟩ ⊨ **|**⟨expression⟩**|** | **<**⟨expression⟩**>** | **[**⟨expression⟩**]** | **{**⟨identifier⟩**}** | \⟨expression⟩\
⟨primary-expr⟩ ⊨ **(**⟨expression⟩**)** | ⟨identifier⟩ | ⟨subprog-call⟩ | ⟨literal⟩ | ⟨structure⟩

1.7.1 Literals

⟨literal⟩ ⊨ ⟨int-lit⟩ | ⟨float-lit⟩ | ⟨string-lit⟩ | ⟨bool-lit⟩ | ⟨char-lit⟩
⟨bool-lit⟩ ⊨ **true** | **false**
⟨string-lit⟩ ⊨ `"`{⟨char⟩}`"`
⟨char-lit⟩ ⊨ `'`⟨char⟩`'`
⟨char⟩ ⊨ `implementation dependent`
⟨int-lit⟩ ⊨ [**-**]⟨digit-seq⟩
⟨float-lit⟩ ⊨ [**-**]⟨digit-seq⟩**.**⟨digit-seq⟩
⟨digit-seq⟩ ⊨ ⟨digit⟩{⟨digit⟩}

Observation A char must be one character, of an enconding defined by the implementation.

1.7.2 Structures

⟨structure⟩ ⊨ ⟨tuple⟩ | ⟨list⟩ | ⟨dict⟩ | ⟨graph⟩ | ⟨user-type⟩ | ⟨edge⟩
⟨tuple⟩ ⊨ **(**⟨expr-list⟩**)**
⟨dict⟩ ⊨ **|**[⟨dict-entry-list⟩]**|**
⟨dict-entry⟩ ⊨ ⟨expression⟩**?**⟨expression⟩
⟨dict-entry-list⟩ ⊨ ⟨dict-entry⟩{**,**⟨dict-entry⟩}
⟨user-type⟩ ⊨ ⟨user-type-id⟩**{**[⟨id-attr-list⟩]**}**
⟨list⟩ ⊨ **[**[⟨list-expr⟩]**]**
⟨list-expr⟩ ⊨ ⟨expr-list⟩ | ⟨list-comprehension⟩
⟨list-comprehension⟩ ⊨ ⟨expression⟩⟨for-loop⟩[⟨comp-condition⟩]
⟨graph-comprehension⟩ ⊨ ⟨edge⟩⟨for-loop⟩[⟨comp-condition⟩]
⟨comp-condition⟩ ⊨ **when (**⟨expression⟩**)**
⟨graph⟩ ⊨ **<**(⟨vertex-set⟩, ⟨edge-set⟩ | ⟨vertex-set⟩ | ⟨edge-set⟩)**>**
⟨vertex-set⟩ ⊨ ⟨expression⟩
⟨edge-set⟩ ⊨ [⟨edge-weight⟩]⟨graph-comprehension⟩
⟨edge-weight⟩ ⊨ ⟨expression⟩ **where**
⟨edge⟩ ⊨ ⟨expression⟩⟨edge-symbol⟩⟨expression⟩
⟨edge-symbol⟩ ⊨ −− | − > | < −

1.7.3 Operators

$$
\begin{aligned}
\langle \text{rel-op} \rangle &\models > \mid < \mid <= \mid >= \\
\langle \text{equality-op} \rangle &\models == \mid \mathbf{!}= \\
\langle \text{unary-op} \rangle &\models + \mid - \\
\langle \text{add-op} \rangle &\models + \mid - \\
\langle \text{mult-op} \rangle &\models * \mid / \mid \% \mid ++ \mid ** \\
\langle \text{exp-op} \rangle &\models \hat{} \\
\end{aligned}
$$